MATHEMATICS OF COMPUTATION
VOLUME 49, NUMBER 180
OCTOBER 1987, PAGES S13-S17

Supplement to A Table of Elliptic Integrals of the Second Kind

By B. C. Carlson

This supplement contains Fortran routines for the standard functions $R_F(x,y,z)$ and $R_D(x,y,z)$, followed by two examples of their use in computing elliptic integrals.

```
C***********************************************************************
C
      DOUBLE PRECISION FUNCTION RF(X,Y,Z,ERRTOL,IERR)
C
C          THIS FUNCTION SUBROUTINE COMPUTES THE INCOMPLETE ELLIPTIC
C          INTEGRAL OF THE FIRST KIND,
C          RF(X,Y,Z) = INTEGRAL FROM ZERO TO INFINITY OF
C
C                                   -1/2      -1/2      -1/2
C                        (1/2)(T+X)    (T+Y)     (T+Z)      DT,
C
C          WHERE X, Y, AND Z ARE NONNEGATIVE AND AT MOST ONE OF THEM
C          IS ZERO.  IF ONE OF THEM IS ZERO, THE INTEGRAL IS COMPLETE.
C          THE DUPLICATION THEOREM IS ITERATED UNTIL THE VARIABLES ARE
C          NEARLY EQUAL, AND THE FUNCTION IS THEN EXPANDED IN TAYLOR
C          SERIES TO FIFTH ORDER.
C          REFERENCES: B. C. CARLSON AND E. M. NOTIS, ALGORITHMS FOR
C          INCOMPLETE ELLIPTIC INTEGRALS, ACM TRANSACTIONS ON MATHEMA-
C          TICAL SOFTWARE, 7 (1981), 398-403; B. C. CARLSON, COMPUTING
C          ELLIPTIC INTEGRALS BY DUPLICATION, NUMER. MATH., 33 (1979),
C          1-16.
C          AUTHORS: B. C. CARLSON AND ELAINE M. NOTIS, AMES LABORATORY-
C          DOE, IOWA STATE UNIVERSITY, AMES, IA 50011, AND R. L. PEXTON,
C          LAWRENCE LIVERMORE NATIONAL LABORATORY, LIVERMORE, CA 94550.
C          AUG. 1, 1979, REVISED JAN. 15, 1987.
C
C          CHECK VALUE: RF(0,1,2) = 1.31102 87771 46059 90523 24198
C          CHECK BY ADDITION THEOREM: RF(X,X+Z,X+W) + RF(Y,Y+Z,Y+W)
C          = RF(0,Z,W), WHERE X,Y,Z,W ARE POSITIVE AND X*Y = Z*W.
C
      INTEGER IERR,PRINTR
      DOUBLE PRECISION C1,C2,C3,E2,E3,EPSLON,ERRTOL,LAMDA
      DOUBLE PRECISION LOLIM,MU,S,UPLIM,X,XN,XNDEV,XNROOT
      DOUBLE PRECISION Y,YN,YNDEV,YNROOT,Z,ZN,ZNDEV,ZNROOT
C          INTRINSIC FUNCTIONS USED: DABS,DMAX1,DMIN1,DSQRT
C
C          PRINTR IS THE UNIT NUMBER OF THE PRINTER.
      DATA PRINTR/6/
C
```


0025-5718/87 $1.00 + $.25 per page

```
C        LOLIM DETERMINES THE LOWER LIMIT AND UPLIM THE UPPER LIMIT
C        OF THE RANGE OF ADMISSIBLE VALUES OF X, Y, AND Z FOR WHICH
C        THE COMPUTATION WILL PROCEED WITHOUT UNDERFLOW OR OVERFLOW.
C        LOLIM IS NOT LESS THAN THE MACHINE MINIMUM MULTIPLIED BY 5.
C        UPLIM IS NOT GREATER THAN THE MACHINE MAXIMUM DIVIDED BY 5.
C
C        ACCEPTABLE VALUES FOR:   LOLIM        UPLIM
C        IBM 360/370 SERIES   :   3.0D-78      1.0D+75
C        CDC 6000/7000 SERIES :   1.0D-292     1.0D+321
C        UNIVAC 1100 SERIES   :   1.0D-307     1.0D+307
C        CRAY                 :   2.3D-2466    1.0D+2465
C        VAX 11 SERIES        :   1.5D-38      3.0D+37
C        IBM PC               :   1.5D-38      3.0D+37
C
C        WARNING: IF THIS PROGRAM IS CONVERTED TO SINGLE PRECISION,
C        THE VALUES FOR THE UNIVAC 1100 SERIES SHOULD BE CHANGED TO
C        LOLIM = 1.E-37 AND UPLIM = 1.E+37 BECAUSE THE MACHINE
C        EXTREMA CHANGE WITH THE PRECISION.
C
      DATA LOLIM/1.5D-38/, UPLIM/3.0D+37/
C
C        ON INPUT:
C
C        X, Y, AND Z ARE THE VARIABLES IN THE INTEGRAL RF(X,Y,Z).
C
C        ERRTOL IS CHOSEN TO DETERMINE THE ACCURACY OF THE COMPUTED
C        APPROXIMATION TO THE INTEGRAL.  TRUNCATION OF A TAYLOR SERIES
C        AFTER TERMS OF FIFTH ORDER INTRODUCES A RELATIVE ERROR LESS
C        THAN THE AMOUNT SHOWN IN THE SECOND COLUMN OF THE FOLLOWING
C        TABLE FOR EACH VALUE OF ERRTOL IN THE FIRST COLUMN.  IN ADDI-
C        TION TO THE TRUNCATION ERROR THERE WILL BE ROUNDOFF ERROR,
C        BUT IN PRACTICE THE TOTAL ERROR FROM BOTH SOURCES IS USUALLY
C        LESS THAN THE AMOUNT GIVEN IN THE TABLE.  SINCE THE TRUNCA-
C        TION ERROR IS LESS THAN (ERRTOL**6)/(4*(1-ERRTOL)), DECREAS-
C        ING ERRTOL BY A FACTOR OF 10 YIELDS SIX MORE DECIMAL DIGITS
C        OF ACCURACY AT THE EXPENSE OF ONE OR TWO MORE ITERATIONS OF
C        THE DUPLICATION THEOREM.
C
C        SAMPLE CHOICES:  ERRTOL    RELATIVE TRUNCATION
C                                   ERROR LESS THAN
C                         1.0D-3    3.0D-19
C                         3.0D-3    2.0D-16
C                         1.0D-2    3.0D-13
C                         3.0D-2    2.0D-10
C                         1.0D-1    3.0D-7
C
C        ON OUTPUT:
C
C        X, Y, Z, AND ERRTOL ARE UNALTERED.
C
C        IERR IS THE RETURN ERROR CODE:
C             IERR = 0 FOR NORMAL COMPLETION OF THE SUBROUTINE,
C             IERR = 1 FOR ABNORMAL TERMINATION.
C        ************************************************************
C        WARNING: CHANGES IN THE PROGRAM MAY IMPROVE SPEED AT THE
C        EXPENSE OF ROBUSTNESS.
C
      IF (DMIN1(X,Y,Z) .LT. 0.D0) GO TO 100
      IF (DMIN1(X+Y,X+Z,Y+Z) .LT. LOLIM) GO TO 100
      IF (DMAX1(X,Y,Z) .LE. UPLIM) GO TO 112
  100 WRITE(PRINTR,104)
  104 FORMAT(1H0,42H*** ERROR - INVALID ARGUMENTS PASSED TO RF)
      WRITE(PRINTR,108) X,Y,Z
  108 FORMAT(1H ,4HX = ,D23.16,4X,4HY = ,D23.16,4X,4HZ = ,D23.16)
      IERR = 1
      GO TO 124
C
  112 IERR = 0
      XN = X
      YN = Y
      ZN = Z
C
  116 MU = (XN + YN + ZN) / 3.D0
      XNDEV = 2.D0 - (MU + XN) / MU
      YNDEV = 2.D0 - (MU + YN) / MU
      ZNDEV = 2.D0 - (MU + ZN) / MU
      EPSLON = DMAX1(DABS(XNDEV),DABS(YNDEV),DABS(ZNDEV))
      IF (EPSLON .LT. ERRTOL) GO TO 120
      XNROOT = DSQRT(XN)
      YNROOT = DSQRT(YN)
      ZNROOT = DSQRT(ZN)
      LAMDA = XNROOT * (YNROOT + ZNROOT) + YNROOT * ZNROOT
      XN = (XN + LAMDA) * 0.25D0
      YN = (YN + LAMDA) * 0.25D0
      ZN = (ZN + LAMDA) * 0.25D0
      GO TO 116
C
  120 C1 = 1.D0 / 24.D0
      C2 = 3.D0 / 44.D0
      C3 = 1.D0 / 14.D0
      E2 = XNDEV * YNDEV - ZNDEV * ZNDEV
      E3 = XNDEV * YNDEV * ZNDEV
      S = 1.D0 + (C1 * E2 - 0.1D0 - C2 * E3) * E2 + C3 * E3
      RF = S / DSQRT(MU)
C
  124 RETURN
      END
```

```
C***********************************************************************
C
      DOUBLE PRECISION FUNCTION RD(X,Y,Z,ERRTOL,IERR)
C
C          THIS FUNCTION SUBROUTINE COMPUTES AN INCOMPLETE ELLIPTIC
C          INTEGRAL OF THE SECOND KIND,
C          RD(X,Y,Z) = INTEGRAL FROM ZERO TO INFINITY OF
C
C                                    -1/2      -1/2      -3/2
C                        (3/2)(T+X)     (T+Y)     (T+Z)     DT,
C
C          WHERE X AND Y ARE NONNEGATIVE, X + Y IS POSITIVE, AND Z IS
C          POSITIVE.  IF X OR Y IS ZERO, THE INTEGRAL IS COMPLETE.
C          THE DUPLICATION THEOREM IS ITERATED UNTIL THE VARIABLES ARE
C          NEARLY EQUAL, AND THE FUNCTION IS THEN EXPANDED IN TAYLOR
C          SERIES TO FIFTH ORDER.
C          REFERENCE: B. C. CARLSON AND E. M. NOTIS, ALGORITHMS FOR
C          INCOMPLETE ELLIPTIC INTEGRALS, ACM TRANSACTIONS ON MATHEMA-
C          TICAL SOFTWARE, 7 (1981), 398-403; B. C. CARLSON, COMPUTING
C          ELLIPTIC INTEGRALS BY DUPLICATION, NUMER. MATH. 33 (1979),
C          1-16.
C          AUTHORS: B. C. CARLSON AND ELAINE M. NOTIS, AMES LABORATORY-
C          DOE, IOWA STATE UNIVERSITY, AMES, IA 50011, AND R. L. PEXTON,
C          LAWRENCE LIVERMORE NATIONAL LABORATORY, LIVERMORE, CA 94550.
C          AUG. 1, 1979, REVISED JAN. 15, 1987.
C
C          CHECK VALUE: RD(0,2,1) = 1.79721 03521 03388 31115 98837
C          CHECK: RD(X,Y,Z) + RD(Y,Z,X) + RD(Z,X,Y) = 3/DSQRT(X*Y*Z),
C          WHERE X, Y, AND Z ARE POSITIVE.
C
      INTEGER IERR,PRINTR
      DOUBLE PRECISION C1,C2,C3,C4,EA,EB,EC,ED,EF,EPSLON,ERRTOL,LAMDA
      DOUBLE PRECISION LOLIM,MU,POWER4,SIGMA,S1,S2,UPLIM,X,XN,XNDEV
      DOUBLE PRECISION XNROOT,Y,YN,YNDEV,YNROOT,Z,ZN,ZNDEV,ZNROOT
C          INTRINSIC FUNCTIONS USED: DABS,DMAX1,DMIN1,DSQRT
C
C          PRINTR IS THE UNIT NUMBER OF THE PRINTER.
      DATA PRINTR/6/
C
C          LOLIM DETERMINES THE LOWER LIMIT AND UPLIM THE UPPER LIMIT
C          OF THE RANGE OF ADMISSIBLE VALUES OF X, Y, AND Z FOR WHICH
C          THE COMPUTATION WILL PROCEED WITHOUT UNDERFLOW OR OVERFLOW.
C          LOLIM IS NOT LESS THAN 2/(MACHINE MAXIMUM)**(2/3).
C          UPLIM IS NOT GREATER THAN (0.1 * ERRTOL / MACHINE
C          MINIMUM) ** (2/3), WHERE ERRTOL IS DESCRIBED BELOW.
C          IN THE FOLLOWING TABLE IT IS ASSUMED THAT ERRTOL WILL
C          NEVER BE CHOSEN SMALLER THAN 1.D-5.
C
C          ACCEPTABLE VALUES FOR:   LOLIM          UPLIM
C          IBM 360/370 SERIES   :   6.0D-51        1.0D+48
C          CDC 6000/7000 SERIES :   5.0D-215       2.0D+191
C          UNIVAC 1100 SERIES   :   1.0D-205       2.0D+201
C          CRAY                 :   3.0D-1644      1.6D+1640
C          VAX 11 SERIES        :   1.0D-25        4.5D+21
C          IBM PC               :   1.0D-25        4.5D+21
C
C          WARNING: IF THIS PROGRAM IS CONVERTED TO SINGLE PRECISION,
C          THE VALUES FOR THE UNIVAC 1100 SERIES SHOULD BE CHANGED TO
C          LOLIM = 1.E-25 AND UPLIM = 2.E+21 BECAUSE THE MACHINE
C          EXTREMA CHANGE WITH THE PRECISION.
C
      DATA LOLIM/1.0D-25/, UPLIM/4.5D+21/
C
C          ON INPUT:
C
C          X, Y, AND Z ARE THE VARIABLES IN THE INTEGRAL RD(X,Y,Z).
C
C          ERRTOL IS CHOSEN TO DETERMINE THE ACCURACY OF THE COMPUTED
C          APPROXIMATION TO THE INTEGRAL.  TRUNCATION OF A TAYLOR SERIES
C          AFTER TERMS OF FIFTH ORDER INTRODUCES A RELATIVE ERROR LESS
C          THAN THE AMOUNT SHOWN IN THE SECOND COLUMN OF THE FOLLOWING
C          TABLE FOR EACH VALUE OF ERRTOL IN THE FIRST COLUMN.  IN ADDI-
C          TION TO THE TRUNCATION ERROR THERE WILL BE ROUNDOFF ERROR,
C          BUT IN PRACTICE THE TOTAL ERROR FROM BOTH SOURCES IS USUALLY
C          LESS THAN THE AMOUNT GIVEN IN THE TABLE.  SINCE THE TRUNCA-
C          TION ERROR IS LESS THAN 3 * ERRTOL ** 6 / (1-ERRTOL) ** 3/2,
C          DECREASING ERRTOL BY A FACTOR OF 10 YIELDS SIX MORE DECIMAL
C          DIGITS OF ACCURACY AT THE EXPENSE OF ONE OR TWO MORE ITERA-
C          TIONS OF THE DUPLICATION THEOREM.
C
C          SAMPLE CHOICES:  ERRTOL    RELATIVE TRUNCATION
C                                     ERROR LESS THAN
C                           1.0D-3    4.0D-18
C                           3.0D-3    3.0D-15
C                           1.0D-2    4.0D-12
C                           3.0D-2    3.0D-9
C                           1.0D-1    4.0D-6
C
C          ON OUTPUT:
C
C          X, Y, Z, AND ERRTOL ARE UNALTERED.
C
C          IERR IS THE RETURN ERROR CODE:
C               IERR = 0 FOR NORMAL COMPLETION OF THE SUBROUTINE,
C               IERR = 1 FOR ABNORMAL TERMINATION.
```

```
C         ***************************************************************
C         WARNING: CHANGES IN THE PROGRAM MAY IMPROVE SPEED AT THE
C         EXPENSE OF ROBUSTNESS.
C
      IF (DMIN1(X,Y) .LT. 0.D0) GO TO 100
      IF (DMIN1(X+Y,Z) .LT. LOLIM) GO TO 100
      IF (DMAX1(X,Y,Z) .LE. UPLIM) GO TO 112
  100 WRITE(PRINTR,104)
  104 FORMAT(1H0,42H*** ERROR - INVALID ARGUMENTS PASSED TO RD)
      WRITE(PRINTR,108) X,Y,Z
  108 FORMAT(1H ,4HX = ,D23.16,4X,4HY = ,D23.16,4X,4HZ = ,D23.16)
      IERR = 1
      GO TO 124
C
  112 IERR = 0
      XN = X
      YN = Y
      ZN = Z
      SIGMA = 0.D0
      POWER4 = 1.D0
C
  116 MU = (XN + YN + 3.D0 * ZN) * 0.2D0
      XNDEV = (MU - XN) / MU
      YNDEV = (MU - YN) / MU
      ZNDEV = (MU - ZN) / MU
      EPSLON = DMAX1(DABS(XNDEV),DABS(YNDEV),DABS(ZNDEV))
      IF (EPSLON .LT. ERRTOL) GO TO 120
      XNROOT = DSQRT(XN)
      YNROOT = DSQRT(YN)
      ZNROOT = DSQRT(ZN)
      LAMDA = XNROOT * (YNROOT + ZNROOT) + YNROOT * ZNROOT
      SIGMA = SIGMA + POWER4 / (ZNROOT * (ZN + LAMDA))
      POWER4 = POWER4 * 0.25D0
      XN = (XN + LAMDA) * 0.25D0
      YN = (YN + LAMDA) * 0.25D0
      ZN = (ZN + LAMDA) * 0.25D0
      GO TO 116
C
  120 C1 = 3.D0 / 14.D0
      C2 = 1.D0 / 6.D0
      C3 = 9.D0 / 22.D0
      C4 = 3.D0 / 26.D0
      EA = XNDEV * YNDEV
      EB = ZNDEV * ZNDEV
      EC = EA - EB
      ED = EA - 6.D0 * EB
      EF = ED + EC + EC
      S1 = ED * (- C1 + 0.25D0 * C3 * ED - 1.5D0 * C4 * ZNDEV * EF)
      S2 = ZNDEV * (C2 * EF + ZNDEV * (- C3 * EC + ZNDEV * C4 * EA))
      RD = 3.D0 * SIGMA + POWER4 * (1.D0 + S1 + S2) / (MU * DSQRT(MU))
C
  124 RETURN
      END
```

Example 1. The arc of Bernoulli's lemniscate $r^2 = \cos(2\theta)$ between the points (r,θ) and (ρ,ϕ), where $0 \le \theta \le \phi \le \pi/4$ and $0 \le \rho \le r \le 1$, has length [2,Ex.8.3-7]

$$s = \int_\rho^r (1-u^4)^{-1/2}du. \tag{S.1}$$

Putting $u^2 = t$ and using (2.6) and (2.1)-(2.4), we find

$$s = \frac{1}{2}\int_{\rho^2}^{r^2} [t(1+t)(1-t)]^{-1/2}dt = R_F(U_{12}^2, U_{13}^2, U_{14}^2),$$

$$(r^2-\rho^2)U_{14} = r(1-\rho^4)^{1/2} + \rho(1-r^4)^{1/2}, \tag{S.2}$$

$$U_{12}^2 = U_{14}^2 + d_{13}d_{24} = U_{14}^2 + (-1)(-1) = U_{14}^2 + 1,$$

$$U_{13}^2 = U_{14}^2 + d_{12}d_{34} = U_{14}^2 + (-1)(1) = U_{14}^2 - 1.$$

We have chosen $a_1+b_1t = t$, $a_2+b_2t = 1+t$, $a_3+b_3t = 1-t$, and $a_4+b_4t = 1$. If $\rho = 1/\sqrt{3}$ and $r = 1/\sqrt{2}$, then $U_{14} = 7$ and

$$s_1 = \int_{1/\sqrt{3}}^{1/\sqrt{2}} (1-u^4)^{-1/2}du = R_F(48,49,50) = 0.14286\ 30937\ 9176\ldots. \tag{S.3}$$

Here we have used the symmetry of R_F and the first Fortran code in this Supplement.

If $\rho = 0$, U_{14} reduces to $1/r$ and the arc length to

$$\int_0^r (1-u^4)^{-1/2}du = R_F(r^{-2}-1, r^{-2}, r^{-2}+1), \tag{S.4}$$

in agreement with [2,Ex.8.3-7]. The case $r = 1$, representing the length of a quadrant of the lemniscate [8],[10], is $R_F(0,1,2)$, with numerical value given to 25D in the comments of the Fortran code. A check on (S.3) is provided by splitting the integral into two parts,

$$s_2 = \int_0^{1/\sqrt{2}} (1-u^4)^{-1/2} du = R_F(1,2,3) = 0.72694\ 59354\ 6891\ldots, \tag{S.5}$$

$$s_3 = \int_0^{1/\sqrt{3}} (1-u^4)^{-1/2} du = R_F(2,3,4) = 0.58408\ 28416\ 7715\ldots. \tag{S.6}$$

We see that $s_1 = s_2 - s_3$ to 14D. The relation

$$R_F(1,2,3) - R_F(2,3,4) = R_F(48,49,50) \tag{S.7}$$

is a special case of the addition theorem (4.7).

Example 2. With the same notation and procedure as in Example 1, we find from (2.7) that

$$I = \int_\rho^r u^2(1-u^4)^{-1/2} du = \frac{1}{3} R_D(U_{14}^2-1, U_{14}^2+1, U_{14}^2) + r\rho/U_{14}, \tag{S.8}$$

where U_{14} is given in (S.2). A special case is

$$I_1 = \int_{1/\sqrt{3}}^{1/\sqrt{2}} u^2(1-u^4)^{-1/2} du = \frac{1}{3} R_D(48,50,49) + 1/7\sqrt{6} \tag{S.9}$$

$$= \frac{1}{3} \times 0.00291\ 57121\ 46567\ 96\ldots + 0.05832\ 11843\ 51980\ 43\ldots$$

$$= 0.05929\ 30884\ 00836\ 4\ldots,$$

where we have used the second Fortran code in this Supplement.

If $\rho = 0$, (S.8) reduces to

$$\int_0^r u^2(1-u^4)^{-1/2} du = \frac{1}{3} R_D(r^{-2}-1, r^{-2}+1, r^{-2}), \tag{S.10}$$

and the case $r = 1$ is the second lemniscate constant [8],[10],

$$\int_0^1 u^2(1-u^4)^{-1/2} du = \frac{1}{3} R_D(0,2,1) = 0.59907\ 01173\ 67796\ldots. \tag{S.11}$$

The value of $R_D(0,2,1)$ is given to 25D in the comments of the Fortran code, and a well-known relation between the two lemniscate constants [10,Theorem 2] takes the form $R_F(0,1,2)R_D(0,2,1) = 3\pi/4$. A check on (S.9) is provided by splitting the integral into two parts,

$$I_2 = \int_0^{1/\sqrt{2}} u^2(1-u^4)^{-1/2} du = \frac{1}{3} R_D(1,3,2) = 0.12505\ 74576\ 52385\ldots, \tag{S.12}$$

$$I_3 = \int_0^{1/\sqrt{3}} u^2(1-u^4)^{-1/2} du = \frac{1}{3} R_D(2,4,3) = 0.06576\ 43692\ 51548\ldots. \tag{S.13}$$

We see that $I_1 = I_2 - I_3$ to 13S. The relation

$$R_D(1,3,2) - R_D(2,4,3) = R_D(48,50,49) + \sqrt{6}/14 \tag{S.14}$$

is a special case of the addition theorem (4.13).